NAG Fortran Library Routine Document

D01FCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D01FCF attempts to evaluate a multi-dimensional integral (up to 15 dimensions), with constant and finite limits, to a specified relative accuracy, using an adaptive subdivision strategy.

2 Specification

```
 SUBROUTINE D01FCF(NDIM, A, B, MINPTS, MAXPTS, FUNCTN, EPS, ACC, LENWRK,
1                  WRKSTR, FINVAL, IFAIL)
 INTEGER           NDIM, MINPTS, MAXPTS, LENWRK, IFAIL
 real              A(NDIM), B(NDIM), FUNCTN, EPS, ACC, WRKSTR(LENWRK),
1                  FINVAL
 EXTERNAL          FUNCTN
```

3 Description

The routine returns an estimate of a multi-dimensional integral over a hyper-rectangle (i.e., with constant limits), and also an estimate of the relative error. The user sets the relative accuracy required, supplies the integrand as a function subprogram (FUNCTN), and also sets the minimum and maximum acceptable number of calls to FUNCTN (in MINPTS and MAXPTS).

The routine operates by repeated subdivision of the hyper-rectangular region into smaller hyper-rectangles. In each subregion, the integral is estimated using a seventh-degree rule, and an error estimate is obtained by comparison with a fifth-degree rule which uses a subset of the same points. The fourth differences of the integrand along each co-ordinate axis are evaluated, and the subregion is marked for possible future subdivision in half along that co-ordinate axis which has the largest absolute fourth difference.

If the estimated errors, totalled over the subregions, exceed the requested relative error (or if fewer than MINPTS calls to FUNCTN have been made), further subdivision is necessary, and is performed on the subregion with the largest estimated error, that subregion being halved along the appropriate co-ordinate axis.

The routine will fail if the requested relative error level has not been attained by the time MAXPTS calls to FUNCTN have been made; or, if the amount LENWRK of working storage is insufficient. A formula for the recommended value of LENWRK is given in Section 5. If a smaller value is used, and is exhausted in the course of execution, the routine switches to a less efficient mode of operation; only if this mode also breaks down is insufficient storage reported.

D01FCF is based on the HALF subroutine developed by van Dooren and De Ridder (1976). It uses a different basic rule, described by Genz and Malik (1980).

4 References

van Dooren P and De Ridder L (1976) An adaptive algorithm for numerical integration over an N-dimensional cube *J. Comput. Appl. Math.* **2** 207–217

Genz A C and Malik A A (1980) An Adaptive Algorithm for Numerical Integration over an N-dimensional Rectangular Region *J. Comput. Appl. Math.* **6** 295–302

5 Parameters

1: NDIM – INTEGER *Input*

On entry: the number of dimensions of the integral, n.

Constraint: $2 \leq \text{NDIM} \leq 15$.

2: A(NDIM) – ***real*** array *Input*

On entry: the lower limits of integration, a_i, for $i = 1, 2, \ldots, n$.

3: B(NDIM) – ***real*** array *Input*

On entry: the upper limits of integration, b_i, for $i = 1, 2, \ldots, n$.

4: MINPTS – INTEGER *Input/Output*

On entry: MINPTS must be set to the minimum number of integrand evaluations to be allowed.

On exit: MINPTS contains the actual number of integrand evaluations used by D01FCF.

5: MAXPTS – INTEGER *Input*

On entry: the maximum number of integrand evaluations to be allowed.

Constraints:

MAXPTS $\geq$ MINPTS
MAXPTS $\geq \alpha$,
where $\alpha = 2^{\text{NDIM}} + 2 \times \text{NDIM}^2 + 2 \times \text{NDIM} + 1$.

6: FUNCTN – ***real*** FUNCTION, supplied by the user. *External Procedure*

FUNCTN must return the value of the integrand f at a given point.

Its specification is:

```
real FUNCTION FUNCTN(NDIM, Z)
INTEGER              NDIM
real                 Z(NDIM)
```

1: NDIM – INTEGER *Input*

On entry: the number of dimensions of the integral, n.

2: Z(NDIM) – ***real*** array *Input*

On entry: the co-ordinates of the point at which the integrand must be evaluated.

FUNCTN must be declared as EXTERNAL in the (sub)program from which D01FCF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

7: EPS – ***real*** *Input*

On entry: the relative error acceptable to the user. When the solution is zero or very small relative accuracy may not be achievable but the user may still set EPS to a reasonable value and check for the error exit IFAIL = 2.

Constraint: EPS > 0.0.

8: ACC – ***real*** *Output*

On exit: the estimated relative error in FINVAL.

9: LENWRK – INTEGER *Input*

On entry: the dimension of the array WRKSTR as declared in the (sub)program from which D01FCF is called.

Suggested value: for maximum efficiency, LENWRK $\geq$ (NDIM + 2) $\times$ (1 + MAXPTS/α) (see parameter MAXPTS for α).

If LENWRK is less than this, the routine will usually run less efficiently and may fail.

Constraint: LENWRK $\geq 2 \times$ NDIM + 4.

10: WRKSTR(LENWRK) – ***real*** array *Workspace*

11: FINVAL – ***real*** *Output*

On exit: the best estimate obtained for the integral.

12: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq$ 0 on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NDIM $<$ 2,
or NDIM $>$ 15,
or MAXPTS is too small,
or LENWRK $< 2 \times$ NDIM + 4,
or EPS $\leq$ 0.0.

IFAIL = 2

MAXPTS was too small to obtain the required relative accuracy EPS. On soft failure, FINVAL and ACC contain estimates of the integral and the relative error, but ACC will be greater than EPS.

IFAIL = 3

LENWRK was too small. On soft failure, FINVAL and ACC contain estimates of the integral and the relative error, but ACC will be greater than EPS.

7 Accuracy

A relative error estimate is output through the parameter ACC.

8 Further Comments

Execution time will usually be dominated by the time taken to evaluate the integrand FUNCTN, and hence the maximum time that could be taken will be proportional to MAXPTS.

9 Example

This example program estimates the integral

$$\int_0^1 \int_0^1 \int_0^1 \int_0^1 \frac{4z_1 z_3^2 \exp(2z_1 z_3)}{(1+z_2+z_4)^2}\, dz_4\, dz_3\, dz_2\, dz_1 = 0.575364.$$

The accuracy requested is one part in 10,000.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     D01FCF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NDIM, MAXPTS, LENWRK
      PARAMETER        (NDIM=4,MAXPTS=1000*NDIM,LENWRK=(NDIM+2)
     +                 *(1+MAXPTS/(2**NDIM+2*NDIM*NDIM+2*NDIM+1)))
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      real             ACC, EPS, FINVAL
      INTEGER          IFAIL, K, MINPTS
*     .. Local Arrays ..
      real             A(NDIM), B(NDIM), WRKSTR(LENWRK)
*     .. External Functions ..
      real             FUNCTN
      EXTERNAL         FUNCTN
*     .. External Subroutines ..
      EXTERNAL         D01FCF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D01FCF Example Program Results'
      DO 20 K = 1, NDIM
         A(K) = 0.0e0
         B(K) = 1.0e0
   20 CONTINUE
      EPS = 0.0001e0
      MINPTS = 0
      IFAIL = 1
*
      CALL D01FCF(NDIM,A,B,MINPTS,MAXPTS,FUNCTN,EPS,ACC,LENWRK,WRKSTR,
     +            FINVAL,IFAIL)
*
      WRITE (NOUT,*)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99999) 'IFAIL =', IFAIL
         WRITE (NOUT,*)
      END IF
      IF (IFAIL.EQ.0 .OR. IFAIL.GE.2) THEN
         WRITE (NOUT,99998) 'Requested accuracy = ', EPS
         WRITE (NOUT,99997) 'Estimated value    = ', FINVAL
         WRITE (NOUT,99998) 'Estimated accuracy = ', ACC
      END IF
      STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,A,e12.2)
99997 FORMAT (1X,A,F12.4)
      END
*
      real FUNCTION FUNCTN(NDIM,Z)
*     .. Scalar Arguments ..
      INTEGER              NDIM
*     .. Array Arguments ..
      real                 Z(NDIM)
*     .. Intrinsic Functions ..
      INTRINSIC            EXP
```

```
*     .. Executable Statements ..
      FUNCTN = 4.0e0*Z(1)*Z(3)*Z(3)*EXP(2.0e0*Z(1)*Z(3))/(1.0e0+Z(2)
     +         +Z(4))**2
      RETURN
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 D01FCF Example Program Results

 Requested accuracy =     0.10E-03
 Estimated value    =       0.5754
 Estimated accuracy =     0.99E-04
```